HARALD NAHRSTEDT

Excel + VBA

Ergänzungen

Kapitel

Einführung in VBA

Sequentielle Textdateien

Erstellt am 12.02.2012

Beschreibung

In Textdateien lassen sich schnell Informationen speichern und auch wieder schnell abrufen. Dabei sind diese Dateien für ein Schreiben und Lesen des gesamten Textes gedacht. Eben ein sequentielles Schreiben und Lesen von Textzeilen. Andere Methoden lassen sich mit einem kleinen Programmieraufwand auch bewerkstelligen. Die Handhabung der Daten geht bei großen Datenmengen aber deutlich nur mit großem Zeitaufwand.

1 Struktur einer sequentiellen Datei

In Textdateien lassen sich schnell Informationen speichern und wieder abrufen. Dies trifft nur für den gesamten Inhalt einer Textdatei zu. Das liegt daran, dass die Zeilen einer Textdatei nur sequentiell hintereinander in eine Textdatei geschrieben, bzw. gelesen werden können. Daher auch die Bezeichnung *sequentielle Datei*. Eine sequentielle Datei hat folgende Struktur

1. Zeile: Eine beliebige Folge von Textzeichen CR LF
2. Zeile: Eine beliebige Folge von Textzeichen CR LF
3. …

In einer Zeile steht eine beliebige Anzahl Buchstaben, Ziffern und Sonderzeichen. Die Zeile endet mit zwei Steuerzeichen. Nämlich CR für Carriage Return (VBA-Konstante vbCrb, ASCII-Wert = 13) und LF für Line Feed (VBA-Konstante vbLF, ASCII-Wert = 10). Beide Steuerzeichen sind in der VBA-Konstanten vbCRLF vereint. Die Bezeichnungen stammen noch aus einer Zeit, als jedermann eine mechanische Schreibmaschine benutzte. Hatte man eine Zeile geschrieben und wollte man eine neue Zeile beginnen, dann musste man einen Hebel nach rechts schieben, dadurch wurde der Wagen (am Hebel eine Anschlagwalze mit eingespannten Blatt) nach rechts geschoben (Carriage Return) , also an den Zeilenanfang, und gleichzeitig wurde die Walze mit dem eingespannten Blatt um eine Zeile nach oben gedreht, also eine neue Zeile begonnen (Line Feed). In Word lässt sich ein Modus einstellen, der diese Steuerzeichen anzeigt.

2 In Textdateien schreiben und lesen

Die übliche Handhabung einer Textdatei sieht folgende Vorgehensweise vor. Den gesamten Text einer sequentiellen Datei in eine Variable lesen, Änderungen durchführen und dann den gesamten Text wieder speichern.

Die nachfolgende Prozedur schreibt einen aus mehreren Zeilen bestehenden Text (in einer Variablen) in eine sequentielle Datei. Damit der Inhalt einer bestehenden Datei gleichen Namens verloren geht, wird die Datei im *Output-Modus* geöffnet. Eine weitere Prozedur liest den Inhalt der so erstellten Datei. Wir erstellen ein Modul, das die Prozeduren und Daten einer Sequentiellen Datei enthält.



```
'Modul: mod_01_TextDatei
Option Explicit

'Text in Datei speichern, alter inhalt geht verloren
Public Sub WriteAllText(ByVal sFilename As String, _
   ByVal sLines As String)
   Dim iFile As Integer

   iFile = FreeFile
   Open sFilename For Output As #iFile
   Print #iFile, sLines
   Close #iFile
End Sub

'Lesen des gesamten Inhaltes einer Textdatei
Public Function sReadAllText(ByVal sFilename As String) _
   As String
   Dim iFile   As Integer

   'Existiert die Datei ?
   If Dir(sFilename, vbNormal) <> "" Then
      'Textdatei im Binärmodus öffnen und gesamten
      'Inhalt in einem Rutsch auslesen
      iFile = FreeFile
      Open sFilename For Binary As #iFile
      sReadAllText = Space$(LOF(iFile))
      Get #iFile, , sReadAllText
      Close #iFile
   End If
End Function
```

Die Prozeduren zum Testen der Datei-Prozeduren schreiben wir in ein separates Code-Modul.

```
'Modul: mod_02_TextHandling
Option Explicit

Sub TextDateiSchreiben()
   Dim sFilename As String
   Dim sLines As String

   sLines = "Dies ist ein Testtext!" & vbCrLf
   sLines = sLines & "Dies ist die zweite Zeile." & vbCrLf
   sLines = sLines & "Dies ist die dritte Zeile." & vbCrLf
   sLines = sLines & "Dies ist die vierte Zeile." & vbCrLf
   sLines = sLines & "Dies ist die fünfte Zeile." & vbCrLf
   sLines = sLines & "Dies ist die letzte Zeile."
   sFilename = "C:\Temp\Text\Test.txt"
   Call WriteAllText(sFilename, sLines)
End Sub

Sub TextDateiLesen()
   Dim sFilename As String
```

```
    sFilename = "C:\Temp\Text\Test.txt"
    MsgBox sReadAllText(sFilename)
End Sub
```

Nach Ausführung hat die Datei folgenden Inhalt (Text markiert).



3 Textdateien Text anfügen

Einer bestehenden Textdatei weitere Zeilen anzuhängen, ist im Dateimodus *Append* sehr einfach. Die nachfolgende Datei-Prozedur macht es möglich.

```
'Modul: mod_01_TextDatei

'Zeilen an Textdatei anhängen
Public Sub AppendLineText(ByVal sFilename As String, _
    ByVal sLines As String)
    Dim iFile As Integer

    iFile = FreeFile
    Open sFilename For Append As #iFile
    Print #iFile, sLines
    Close #iFile
End Sub
```

Die entsprechende Testprozedur fügt zwei weitere Zeilen an. Zu beachten ist, dass vbCrLf nur zwischen die Zeilen gesetzt werden muss.

```
'Modul: mod_02_TextHandling

Sub TextDateiErweitern()
    Dim sFilename As String
    Dim sLines As String

    sLines = "Dies ist die erste neue Zeile." & vbCrLf
    sLines = sLines & "Dies ist die zweite neue Zeile."
    sFilename = "C:\Temp\Text\Test.txt"
    Call AppendLineText(sFilename, sLines)
End Sub
```

4 Bestimmte Zeilen aus Textdateien lesen

Möchte man eine bestimmte Zeile aus der Textdatei verwenden, so muss die nachfolgende Datei-Prozedur *sReadLineText* verwendet werden. Sie liest alle Zeilen der Textdatei und zählt sie, bis die entsprechende Zeile gelesen wird.

```
'Modul: mod_01_TextDatei

' Lesen einer bestimmten Zeile einer Textdatei
Public Function sReadLineText(ByVal sFilename As String, _
   ByVal LineToRead As Long) As String
   Dim iFile   As Integer
   Dim sLine   As String
   Dim lRow    As Long

   lRow = 0
   'Existiert die Datei ?
   If Dir(sFilename) <> "" Then
      iFile = FreeFile
      Open sFilename For Input As #iFile
      'Solange einlesen, bis entweder Dateiende
      'oder gewünschte Zeilennummer erreicht
      While Not EOF(iFile) And lRow < LineToRead
         lRow = lRow + 1
         Line Input #iFile, sReadLineText
      Wend
      If EOF(iFile) Then sReadLineText = ""
      Close #iFile
   End If
End Function
```

Die nachfolgende Prozedur testet diese Funktion.

```
'Modul: mod_02_TextHandling

Sub TextZeileLesen()
   Dim sFilename As String

   sFilename = "C:\Temp\Text\Test.txt"
   MsgBox sReadLineText(sFilename, 5)
End Sub
```

5 Textdateien ändern

Etwas komplizierter wird es, will man eine Zeile ersetzen oder einfügen. In diesem Fall gibt es mehrere Möglichkeiten. Wenn die Datei noch nicht existiert, muss diese und die Zeilen vor der Position als Leerzeilen angelegt werden. Existiert die Datei, aber die Position liegt hinter dem Dateiende, dann müssen ebenfalls Leerzeilen bis zur Einfügeposition angelegt werden. Ist die Zeile in der Datei vorhanden, kann sie durch die neue Zeile ersetzt oder hinter der neuen Zeile eingefügt werden.

Dabei wird eine temporäre Datei angelegt, in welcher der neue Inhalt der Quelldatei erstellt wird. Erst danach wird die Quelldatei gelöscht und anschließend durch eine Kopie der temporären Datei ersetzt.

Windows verwaltet ein Verzeichnis für temporäre Dateien und vergibt auch bei Bedarf Namen für temporäre Dateien. Die nachfolgenden Definitionen ergänzen das TextDatei-Modul.

```
'Modul: mod_01_TextDatei

'Ermittelt temporäres Verzeichnis
Private Declare Function GetTempPath Lib "kernel32.dll" _
  Alias "GetTempPathA" ( _
  ByVal nBufferLength As Long, _
  ByVal lpBuffer As String) As Long

'Ermittelt temporären Dateinamen
Private Declare Function GetTempFileName Lib "kernel32" _
  Alias "GetTempFileNameA" ( _
  ByVal lpszPath As String, _
  ByVal lpPrefixString As String, _
  ByVal wUnique As Long, _
  ByVal lpTempFileName As String) As Long

Private sTempPath As String

'Bestimmt temporäre Datei
Public Function sTempFilename() As String
   Dim sTempName As String
   Dim RetVal As Long

   'Temp. Verzeichnis
   If sTempPath = "" Then
      sTempPath = Space$(256)
      RetVal = GetTempPath(Len(sTempPath), sTempPath)
      sTempPath = Left$(sTempPath, RetVal)
   End If

   'Temp. Dateiname
   sTempName = Space$(256)
   Call GetTempFileName(sTempPath, "txt", 0&, sTempName)
   sTempName = Left$(sTempName, _
   InStr(sTempName, Chr$(0)) - 1)
   sTempFilename = sTempName
End Function
```

Auch diese Prozeduren sollen natürlich getestet werden.

Die nachfolgend einfache Testprozedur liefert eine temporäre Datei mit Verzeichnis.

```
'Modul: mod_02_TextHandling

Sub TempDateiBestimmen()
   Dim sText As String
```

```
   sText = sTempFilename
   MsgBox sText
End Sub
```

Mithilfe der nun vorhandenen Datei-Prozeduren kann die gewünschte Datei-Prozedur zum Einfügen oder Ersetzen eine Textzeile in einer sequentiellen Datei erstellt werden.

```
'Modul: mod_01_TextDatei

'Einzelne Zeile in eine Textdatei speichern
Public Sub WriteLineText(ByVal sFilename As String, _
   ByVal LinePos As Long, ByVal sNewLine As String)
   Dim iFile      As Integer
   Dim iTemp      As Integer
   Dim lCount     As Long
   Dim lRow       As Long
   Dim sLine      As String
   Dim sTempFile  As String

   If Dir(sFilename, vbNormal) = "" Then
      'Datei existiert nicht
      iFile = FreeFile
      Open sFilename For Output As #iFile
      'vorherige Leerzeilen erzeugen
      For lCount = 1 To LinePos - 1
         Print #iFile, ""
      Next lCount
      'Zeile speichern
      Print #iFile, sLine
      Close #iFile
   Else
      'Temporäre Datei erstellen
      sTempFile = sTempFilename()
      'Quelldatei zum Lesen und
      'temporäre Datei zum Schreiben öffnen
      iFile = FreeFile: Open sFilename For Input As #iFile
      iTemp = FreeFile: Open sTempFile For Output As #iTemp

      'Original-Datei einlesen und x. Zeile durch
      'neuen Inhalt ersetzen
      lRow = 0
      Do While Not EOF(iFile)
         lRow = lRow + 1
         Line Input #iFile, sLine
         If lRow = LinePos Then
            ' x. Zeile durch neuen Inhalt ersetzen
            sLine = sNewLine
         End If
         Print #iTemp, sLine
      Loop

      'notfalls müssen Leerzeilen ergänzt werden
      If lRow < LinePos Then
```

```
         For lCount = lRow + 1 To LinePos - 1
            Print #iTemp, ""
         Next lCount
         Print #iTemp, sNewLine
      End If

      'Dateien schliessen
      Close #iFile
      Close #iTemp
      'Alte Datei löschen
      Kill sFilename
      'temporäre Datei kopieren
      FileCopy sTempFile, sFilename
      'temporäre Datei löschen
      Kill sTempFile
   End If
End Sub
```

Und auch dafür wird eine Testprozedur benötigt.

```
'Modul: mod_02_TextHandling

Sub TextZeileEinfügen()
   Dim sFilename  As String
   Dim sLine      As String

   sFilename = "C:\Temp\Text\TestNeu.txt"
   sLine = "Dies ist eine eingefügte Zeile."
   Call WriteLineText(sFilename, 5, sLine)
End Sub
```